Computational Logic
A “Hands-on” Introduction to Pure Logic Programming
Syntax: Terms (Variables, Constants, and Structures)

(using Prolog notation conventions)

- **Variables**: start with uppercase character (or “_”), may include “_” and digits:
  
  Examples: X, Im4u, A_little_garden, _, _x, _22

- **Constants**: lowercase first character, may include “_” and digits. Also, numbers and some special characters. Quoted, any character:
  
  Examples: a, dog, a_big_cat, 23, ’Hungry man’, []

- **Structures**: a functor (the structure name, is like a constant name) followed by a fixed number of arguments between parentheses:
  
  Example: date(monday, Month, 1994)

  Arguments can in turn be variables, constants and structures.

  ◊ **Arity**: is the number of arguments of a structure. Functors are represented as *name/arity*. A constant can be seen as a structure with arity zero.

Variables, constants, and structures as a whole are called **terms** (they are the terms of a “first–order language”): the *data structures* of a logic program.
Syntax: Terms

(Using Prolog notation conventions)

- **Examples of terms:**

<table>
<thead>
<tr>
<th>Term</th>
<th>Type</th>
<th>Main functor:</th>
</tr>
</thead>
<tbody>
<tr>
<td>dad</td>
<td>constant</td>
<td>dad/0</td>
</tr>
<tr>
<td>time(min, sec)</td>
<td>structure</td>
<td>time/2</td>
</tr>
<tr>
<td>pair(Calvin, tiger(Hobbes))</td>
<td>structure</td>
<td>pair/2</td>
</tr>
<tr>
<td>Tee(Alf, rob)</td>
<td>illegal</td>
<td>—</td>
</tr>
<tr>
<td>A.good.time</td>
<td>variable</td>
<td>—</td>
</tr>
</tbody>
</table>

- A variable is **free** if it has not been assigned a value yet.
- A term is **ground** if it contains no free variables.
- **Functors** can be defined as *prefix*, *postfix*, or *infix* [operators](#) (just syntax!):

<table>
<thead>
<tr>
<th>a + b</th>
<th>is the term</th>
<th>'+(a,b)</th>
<th>if +/-2 declared infix</th>
</tr>
</thead>
<tbody>
<tr>
<td>- b</td>
<td>is the term</td>
<td>'-'(b)</td>
<td>if -/1 declared prefix</td>
</tr>
<tr>
<td>a &lt; b</td>
<td>is the term</td>
<td>'&lt;'(a,b)</td>
<td>if &lt;/2 declared infix</td>
</tr>
<tr>
<td>john father mary</td>
<td>is the term</td>
<td>father(john,mary)</td>
<td>if father/2 declared infix</td>
</tr>
</tbody>
</table>

We assume that some such operator definitions are always preloaded.
Syntax: Rules and Facts (Clauses)

- **Rule:** an expression of the form:

  \[ p_0(t_1, t_2, \ldots, t_{n_0}) :\neg \ p_1(t_1^1, t_2^1, \ldots, t_{n_1}^1), \]
  
  \[ \ldots \]
  
  \[ p_m(t_1^m, t_2^m, \ldots, t_{n_m}^m). \]

  - \( p_0(\ldots) \) to \( p_m(\ldots) \) are *syntactically* like *terms*.
  - \( p_0(\ldots) \) is called the **head** of the rule.
  - The \( p_i \) to the right of the arrow are called *literals* and form the **body** of the rule. They are also called **procedure calls**.
  - Usually, \( :\neg \) is called the **neck** of the rule.

- **Fact:** an expression of the form \( p(t_1, t_2, \ldots, t_n) \) (i.e., a rule with empty body).

**Example:**

<table>
<thead>
<tr>
<th>Rule</th>
<th>Fact</th>
</tr>
</thead>
<tbody>
<tr>
<td>meal(First, Second, Third) :\neg appetizer(First), main_dish(Second), dessert(Third).</td>
<td>meal(soup, beef, coffee).</td>
</tr>
</tbody>
</table>

- Rules and facts are both called **clauses**.
• **Predicate** (or *procedure definition*): a set of clauses whose heads have the same name and arity (called the **predicate name**).

*Examples:*

```
pet(spot).
pet(X) :- animal(X), barks(X).
pet(X) :- animal(X), meows(X).
```

Predicate `pet/1` has three clauses. Of those, one is a fact and two are rules. Predicate `animal/1` has three clauses, all facts.

• **Logic Program**: a set of predicates.

• **Query**: an expression of the form:

```
?-p_1(t_1^{l_1},...,t_{n_1}^{l_1}),...,p_n(t_1^{n},...,t_{n_m}^{n}).
```

(i.e., a clause without a head).

A query represents a **question to the program**.

*Example:*  
```
?- pet(X).
```
“Declarative” Meaning of Facts and Rules

The declarative meaning is the corresponding one in first order logic, according to certain conventions:

- **Facts**: state things that are true.
  (Note that a fact “p.” can be seen as the rule “p :- true.”)
  
  *Example*: the fact `animal(spot)`.
  can be read as “spot is an animal”.

- **Rules**:
  - Commas in rule bodies represent conjunction, and “:-” represents logical implication (backwards, i.e., if).
  - i.e., `p:-p_1, \cdots, p_m` represents `p \leftarrow p_1 \land \cdots \land p_m`.

  Thus, a rule `p:-p_1, \cdots, p_m` means “if `p_1` and \ldots and `p_m` are true, then `p` is true”

  *Example*: the rule `pet(X):-animal(X), barks(X)`.
  can be read as “X is a pet if it is an animal and it barks”.

- **Variables in facts and rules are universally quantified, \forall (recall clausal form!)**.
“Declarative” Meaning of Predicates and Queries

- **Predicates**: clauses in the same predicate
  
  \[
  p :- p_1, \ldots, p_n \\
  p :- q_1, \ldots, q_m \\
  \ldots
  \]

  provide different *alternatives* (for \( p \)).

  *Example*: the rules
  
  ```prolog
  pet(X) :- animal(X), barks(X).
  pet(X) :- animal(X), meows(X).
  ```

  express two ways for \( X \) to be a pet.

- **Note** (variable *scope*): the \( X \) vars. in the two clauses above are different, despite the same name. Vars. are *local to clauses* (and are *renamed* any time a clause is used—as with vars. local to a procedure in conventional languages).

- **A query** represents a *question to the program*.

  *Examples*:
  
  ```prolog
  ?- pet(spot).
  ?- pet(X).
  ```

  Asks: Is spot a pet?  Asks: “Is there an \( X \) which is a pet?”
"Execution" and Semantics

- **Example of a logic program:**

  ```prolog
  pet(X) :- animal(X), barks(X).
  pet(X) :- animal(X), meows(X).
  animal(tim).   barks(spot).
  animal(spot).  meows(tim).
  animal(hobbes). roars(hobbes).
  ```

- **Execution:** given a program and a query, *executing* the logic program is attempting to find an answer to the query.

  *Example:* given the program above and the query `?- pet(X).` the system will try to find a “substitution” for `X` which makes `pet(X)` true.

  - The **declarative semantics** specifies *what* should be computed (all possible answers).
    - Intuitively, we have two possible answers: `X = spot` and `X = tim`.
  - The **operational semantics** specifies *how* answers are computed (which allows us to determine *how many steps* it will take).
Running Programs in a Logic Programming System

- Create file `pets.pl` with first line (explained later):

```prolog
:- module(_,_,[’bf/bfall’]).
```

+ the pet example code as in the previous slides.

- Interaction with the system query evaluator (the “top level”):

```prolog
Ciao X.Y-...
?- use_module(pets).
yes
?- pet(spot).
yes
?- pet(X).
X = spot ? ;
X = tim ? ;
no
?- 
```

See the part on Developing Programs with a Logic Programming System for more details on the particular system used in the course (Ciao).
Simple (Top-Down) Operational Meaning of Programs

- A logic program is operationally a set of *procedure definitions* (the predicates).
- A query `?- p` is an initial *procedure call*.
- A procedure definition with one *clause* `p :- p_1, \ldots, p_m`. means:
  “to execute a call to `p` you have to *call* `p_1` and \ldots and `p_m`”
    - In principle, the order in which `p_1, \ldots, p_n` are called does not matter, but, in practical systems it is fixed.
- If several clauses (definitions) `p :- p_1, \ldots, p_n` means:
  `p :- q_1, \ldots, q_m`
  “to execute a call to `p`, call `p_1` and \ldots and `p_n`, or, alternatively, `q_1` and \ldots and `q_m`, or \ldots”
    - Unique to logic programming –it is like having several alternative procedure definitions.
    - Means that several possible paths may exist to a solution and they *should be explored*.
    - System usually stops when the first solution found, user can ask for more.
    - Again, in principle, the order in which these paths are explored does not matter (*if certain conditions are met*), but, for a given system, this is typically also fixed.

In the following we define a more precise operational semantics.
Unification: uses

- **Unification** is the mechanism used in *procedure calls* to:
  - Pass parameters.
  - “Return” values.

- It is also used to:
  - Access parts of structures.
  - Give values to variables.

- Unification is a procedure to solve equations on data structures.
  - As usual, it returns a minimal solution to the equation (or the equation system).
  - As many equation solving procedures it is based on isolating variables and then *instantiating* them with their values.
Unification

- **Unifying two terms (or literals) $A$ and $B$:** is asking if they can be made syntactically identical by giving (minimal) values to their variables.
  - I.e., find a **variable substitution** $\theta$ such that $A\theta = B\theta$ (or, if impossible, fail).
  - Only variables can be given values!
  - Two structures can be made identical only by making their arguments identical.

*E.g.:

<table>
<thead>
<tr>
<th>$A$</th>
<th>$B$</th>
<th>$\theta$</th>
<th>$A\theta$</th>
<th>$B\theta$</th>
</tr>
</thead>
<tbody>
<tr>
<td>dog</td>
<td>dog</td>
<td>$\emptyset$</td>
<td>dog</td>
<td>dog</td>
</tr>
<tr>
<td>$X$</td>
<td>a</td>
<td>${X = a}$</td>
<td>a</td>
<td>a</td>
</tr>
<tr>
<td>$X$</td>
<td>$Y$</td>
<td>${X = Y}$</td>
<td>$Y$</td>
<td>$Y$</td>
</tr>
<tr>
<td>$f(X, g(t))$</td>
<td>$f(m(h), g(M))$</td>
<td>${X = m(h), M = t}$</td>
<td>$f(m(h), g(t))$</td>
<td>$f(m(h), g(t))$</td>
</tr>
<tr>
<td>$f(X, g(t))$</td>
<td>$f(m(h), t(M))$</td>
<td>Impossible (1)</td>
<td>$f(m(h), g(t))$</td>
<td>$f(m(h), g(t))$</td>
</tr>
<tr>
<td>$f(X, X)$</td>
<td>$f(Y, l(Y))$</td>
<td>Impossible (2)</td>
<td>$f(Y, l(Y))$</td>
<td>$f(Y, l(Y))$</td>
</tr>
</tbody>
</table>

- (1) Structures with different name and/or arity cannot be unified.
- (2) A variable cannot be given as value a term which contains that variable, because it would create an infinite term. This is known as the **occurs check**. (See, however, **cyclic terms** later.)
Unification

- Often several solutions exist, e.g.:

<p>| | | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>$A$</td>
<td>$B$</td>
<td>$\theta_1$</td>
<td>$A\theta_1$ and $B\theta_1$</td>
</tr>
<tr>
<td>$f(X, g(T))$</td>
<td>$f(m(H), g(M))$</td>
<td>{ $X=m(a)$, $H=a$, $M=b$, $T=b$ }</td>
<td>$f(m(a), g(b))$</td>
</tr>
<tr>
<td>&quot;</td>
<td>&quot;</td>
<td>{ $X=m(H)$, $M=f(A)$, $T=f(A)$ }</td>
<td>$f(m(H), g(f(A)))$</td>
</tr>
</tbody>
</table>

These are correct, but a simpler ("more general") solution exists:

<p>| | | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>$A$</td>
<td>$B$</td>
<td>$\theta_1$</td>
<td>$A\theta_1$ and $B\theta_1$</td>
</tr>
<tr>
<td>$f(X, g(T))$</td>
<td>$f(m(H), g(M))$</td>
<td>{ $X=m(H)$, $T=M$ }</td>
<td>$f(m(H), g(M))$</td>
</tr>
</tbody>
</table>

- Always a unique (modulo variable renaming) *most general* solution exists (unless unification fails).
- This is the one that we are interested in.
- The *unification algorithm* finds this solution.
Unification Algorithm

- Select one equation from the equation system, delete it, and, depending on the form of the equation:
  - $X=X$ : ignore
  - $X=f(\ldots, X, \ldots)$ : fail (occurs check)
  - $X=\text{term}$ :
    - * add it to the solution
    - * replace $X$ by $\text{term}$ anywhere else
  - $a=a$ : ignore
  - $a=b$ : fail
  - $a=f(\ldots)$ : fail
  - $g(\ldots)=f(\ldots)$ : fail
  - $f(\ldots m \ldots)=f(\ldots n \ldots) \ (m \neq n)$ : fail
  - $f(s_1, \ldots, s_n)=f(t_1, \ldots, t_n)$ :
    - * add to the system: $s_1=t_1, \ldots, s_n=t_n$
Unification Algorithm Examples

- Unify: \( p(X, f(b)) \) and \( p(a, Y) \)
  \[
  p(X, f(b)) = p(a, Y) \]
  \[
  X = a \quad Y = f(b)
  \]

- Unify: \( p(X, f(Y)) \) and \( p(a, g(b)) \)
  \[
  p(X, f(Y)) = p(a, g(b)) \]
  \[
  X = a
  
  f(Y) = g(b)
  
  \text{fail}
  \]

- Unify: \( p(X, X) \) and \( p(f(Z), f(W)) \)
  \[
  p(X, X) = p(f(Z), f(W)) \]
  \[
  X = f(Z)
  
  X = f(Z)
  
  X = f(Z)
  
  X = f(W)
  
  X = f(W)
  
  f(Z) = f(W)
  
  Z = W
  
  Z = W
  \]

- Unify: \( p(X, f(Y)) \) and \( p(Z, X) \)
  \[
  p(X, f(Y)) = p(Z, X) \]
  \[
  X = Z
  
  X = Z
  
  X = f(Y)
  
  f(Y) = X
  
  f(Y) = Z
  
  Z = f(Y)
  \]

- Unify: \( p(X, f(X)) \) and \( p(Z, Z) \)
  \[
  p(X, f(X)) = p(Z, Z) \]
  \[
  X = Z
  
  X = Z
  
  X = Z
  
  f(X) = Z \}
  
  f(Z) = Z
  
  \text{fail}
  \]
Unification Algorithm (More Formal Version)

- Let $A$ and $B$ be two terms:

  1. $\theta = \emptyset$, $E = \{A = B\}$
  2. while not $E = \emptyset$:
     2.1 delete an equation $T = S$ from $E$
     2.2 case $T$ or $S$ (or both) are (distinct) variables. Assuming $T$ variable:
        * (occur check) if $T$ occurs in the term $S$ → halt with failure
        * substitute variable $T$ by term $S$ in all terms in $\theta$
        * substitute variable $T$ by term $S$ in all terms in $E$
        * add $T = S$ to $\theta$
     2.3 case $T$ and $S$ are non-variable terms:
        * if their names or arities are different → halt with failure
        * obtain the arguments $\{T_1, \ldots, T_n\}$ of $T$ and $\{S_1, \ldots, S_n\}$ of $S$
        * add $\{T_1 = S_1, \ldots, T_n = S_n\}$ to $E$
  3. halt with $\theta$ being the m.g.u of $A$ and $B$
Unification Algorithm Examples (I)

- Unify: \( A = p(X, X) \) and \( B = p(f(Z), f(W)) \)

<table>
<thead>
<tr>
<th>( \theta )</th>
<th>( E )</th>
<th>( T )</th>
<th>( S )</th>
</tr>
</thead>
<tbody>
<tr>
<td>{}</td>
<td>{ p(X, X) = p(f(Z), f(W)) }</td>
<td>( p(X, X) )</td>
<td>( p(f(Z), f(W)) )</td>
</tr>
<tr>
<td>{}</td>
<td>{ X = f(Z), X = f(W) }</td>
<td>( X )</td>
<td>( f(Z) )</td>
</tr>
<tr>
<td>{ X = f(Z) }</td>
<td>{ f(Z) = f(W) }</td>
<td>( f(Z) )</td>
<td>( f(W) )</td>
</tr>
<tr>
<td>{ X = f(Z) }</td>
<td>{ Z = W }</td>
<td>( Z )</td>
<td>( W )</td>
</tr>
<tr>
<td>{ X = f(W), Z = W }</td>
<td>{}</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

- Unify: \( A = p(X, f(Y)) \) and \( B = p(Z, X) \)

<table>
<thead>
<tr>
<th>( \theta )</th>
<th>( E )</th>
<th>( T )</th>
<th>( S )</th>
</tr>
</thead>
<tbody>
<tr>
<td>{}</td>
<td>{ p(X, f(Y)) = p(Z, X) }</td>
<td>( p(X, f(Y)) )</td>
<td>( p(Z, X) )</td>
</tr>
<tr>
<td>{}</td>
<td>{ X = Z, f(Y) = X }</td>
<td>( X )</td>
<td>( Z )</td>
</tr>
<tr>
<td>{ X = Z }</td>
<td>{ f(Y) = Z }</td>
<td>( f(Y) )</td>
<td>( Z )</td>
</tr>
<tr>
<td>{ X = f(Y), Z = f(Y) }</td>
<td>{}</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Unification Algorithm Examples (II)

- Unify: \( A = p(X, f(Y)) \) and \( B = p(a, g(b)) \)

\[
\begin{array}{cccc}
\theta & E & T & S \\
{} & \{ p(X, f(Y)) = p(a, g(b)) \} & p(X, f(Y)) & p(a, g(b)) \\
{} & \{ X = a, f(Y) = g(b) \} & X & a \\
\{ X = a \} & \{ f(Y) = g(b) \} & f(Y) & g(b) \\
fail & & & \\
\end{array}
\]

- Unify: \( A = p(X, f(X)) \) and \( B = p(Z, Z) \)

\[
\begin{array}{cccc}
\theta & E & T & S \\
{} & \{ p(X, f(X)) = p(Z, Z) \} & p(X, f(X)) & p(Z, Z) \\
{} & \{ X = Z, f(X) = Z \} & X & Z \\
\{ X = Z \} & \{ f(Z) = Z \} & f(Z) & Z \\
fail & & & \\
\end{array}
\]
A (Schematic) Interpreter for Logic Programs (SLD–resolution)

Input: A logic program $P$, a query $Q$
Output: $\mu$ (answer substitution) if $Q$ is provable from $P$, failure otherwise

1. Make a copy $Q'$ of $Q$
2. Initialize the “resolvent” $R$ to be \{Q\}
3. While $R$ is nonempty do:
   3.1. Take a literal $A$ in $R$
   3.2. Take a clause $A':-B_1,\ldots,B_n$ (renamed) from $P$
       with $A'$ same predicate symbol as $A$
       3.2.1. If there is a solution $\theta$ to $A = A'$ (unification)
           • Replace $A$ in $R$ by $B_1,\ldots,B_n$
           • Apply $\theta$ to $R$ and $Q$
       3.2.2. Otherwise, take another clause and repeat
3.3. If there are no more clauses, go back to some other choice
3.4. If there are no pending choices left, output failure
4. ($R$ empty) Output solution $\mu$ to $Q = Q'$
5. Explore another pending branch for more solutions (upon request)
A (Schematic) Interpreter for Logic Programs (Standard Prolog)

Input: A logic program $P$, a query $Q$
Output: $\mu$ (answer substitution) if $Q$ is provable from $P$, failure otherwise

1. Make a copy $Q'$ of $Q$
2. Initialize the “resolvent” $R$ to be $\{Q\}$
3. While $R$ is nonempty do:
   3.1. Take the leftmost literal $A$ in $R$
   3.2. Take the first clause $A':-B_1,\ldots,B_n$ (renamed) from $P$ with $A'$ same predicate symbol as $A$
      3.2.1. If there is a solution $\theta$ to $A = A'$ (unification)
          • Replace $A$ in $R$ by $B_1,\ldots,B_n$
          • Apply $\theta$ to $R$ and $Q$
      3.2.2. Otherwise, take the next clause and repeat
3.3. If there are no more clauses, go back to most recent pending choice
3.4. If there are no pending choices left, output failure
4. ($R$ empty) Output solution $\mu$ to $Q = Q'$
5. Explore the most recent pending branch for more solutions (upon request)
Step 3.2 defines \textit{alternative paths} to be explored to find answer(s); execution explores this tree (for example, breadth-first).

Since step 3.2 is left open, a given logic \textit{programming} system must specify how it deals with this by providing one (or more)

\textbf{Search rule(s)}: “how are clauses/branches selected in 3.2.”

Note that choosing a different clause (in step 3.2) can lead to finding solutions in a different order – Example (two valid executions):

\begin{verbatim}
?- pet(X).
X = spot ? ;
X = tim ? ;
no
?- pet(X).
X = tim ? ;
X = spot ? ;
nono
?- pet(X).
X = spot ? ;
nono
?- pet(X).
X = tim ? ;
X = spot ? ;
no
\end{verbatim}

In fact, there is also some freedom in step 3.1, i.e., a system may also specify:

\textbf{Computation rule(s)}: “how are literals selected in 3.1.”
Running Programs: Alternative Execution Paths

\( C_1: \) pet(X) :- animal(X), barks(X).

\( C_2: \) pet(X) :- animal(X), meows(X).

\( C_3: \) animal(tim).

\( C_4: \) animal(spot).

\( C_5: \) animal(hobbes).

\( C_6: \) barks(spot).

\( C_7: \) meows(tim).

\( C_8: \) roars(hobbes).

\[ \text{• } \bf{?- pet(X).} \text{ (top-down, left-to-right)} \]

<table>
<thead>
<tr>
<th>( Q )</th>
<th>( R )</th>
<th>Clause</th>
<th>( \theta )</th>
</tr>
</thead>
<tbody>
<tr>
<td>pet(X)</td>
<td>pet(X)</td>
<td>( C_1^* )</td>
<td>( { X=X_1 } )</td>
</tr>
<tr>
<td>pet(X_1)</td>
<td>animal(X_1), barks(X_1)</td>
<td>( C_3^* )</td>
<td>( { X_1=\text{tim} } )</td>
</tr>
<tr>
<td>pet(tim)</td>
<td>barks(tim)</td>
<td>???</td>
<td>failure</td>
</tr>
</tbody>
</table>

* means choice-point, i.e., other clauses applicable.

\( \text{• But solutions exist in other paths!} \)
Running Programs: Alternative Execution Paths

\[ C_1: \text{pet}(X) :- \]
\[ \text{animal}(X), \text{barks}(X). \]
\[ C_2: \text{pet}(X) :- \]
\[ \text{animal}(X), \text{meows}(X). \]
\[ C_3: \text{animal}(\text{tim}). \]
\[ C_4: \text{animal}(\text{spot}). \]
\[ C_5: \text{animal}(\text{hobbes}). \]
\[ C_6: \text{barks}(\text{spot}). \]
\[ C_7: \text{meows}(\text{tim}). \]
\[ C_8: \text{roars}(\text{hobbes}). \]

- \( ?- \text{pet}(X). \) (top-down, left-to-right)

<table>
<thead>
<tr>
<th>( Q )</th>
<th>( R )</th>
<th>Clause</th>
<th>( \theta )</th>
</tr>
</thead>
<tbody>
<tr>
<td>\text{pet}(X)</td>
<td>\text{pet}(X)</td>
<td>( C_1^* )</td>
<td>{ \text{X}=\text{X}_1 }</td>
</tr>
<tr>
<td>\text{pet}(\text{X}_1)</td>
<td>\text{animal}(\text{X}_1), \text{barks}(\text{X}_1)</td>
<td>( C_3^* )</td>
<td>{ \text{X}_1=\text{tim} }</td>
</tr>
<tr>
<td>\text{pet}(\text{tim})</td>
<td>\text{barks}(\text{tim})</td>
<td>???</td>
<td>\textit{failure}</td>
</tr>
</tbody>
</table>

* means \textit{choice-point}, i.e., other clauses applicable.

- But solutions exist in other paths!

\[ \rightarrow \text{Let’s go back to our last choice point (}C_3^*\text{) and try the next alternative...} \]
Running Programs: Alternative Execution Paths

$C_1$: \text{pet}(X) :- \\
\quad \text{animal}(X), \text{barks}(X).

$C_2$: \text{pet}(X) :- \\
\quad \text{animal}(X), \text{meows}(X).

$C_3$: \text{animal}(\text{tim}).

$C_4$: \text{animal}(\text{spot}).

$C_5$: \text{animal}(\text{hobbes}).

$C_6$: \text{barks}(\text{spot}).

$C_7$: \text{meows}(\text{tim}).

$C_8$: \text{roars}(\text{hobbes}).

- $\text{?- pet}(X)$. (top-down, left-to-right, different branch)

<table>
<thead>
<tr>
<th>$Q$</th>
<th>$R$</th>
<th>Clause</th>
<th>$\theta$</th>
</tr>
</thead>
<tbody>
<tr>
<td>$\text{pet}(X)$</td>
<td>$\text{pet}(X)$</td>
<td>$C_1^*$</td>
<td>${X=X_1}$</td>
</tr>
<tr>
<td>$\text{pet}(X_1)$</td>
<td>$\text{animal}(X_1), \text{barks}(X_1)$</td>
<td>$C_4^*$</td>
<td>${X_1=\text{spot}}$</td>
</tr>
<tr>
<td>$\text{pet}(\text{spot})$</td>
<td>$\text{barks}(\text{spot})$</td>
<td>$C_6$</td>
<td>${}$</td>
</tr>
<tr>
<td>$\text{pet}(\text{spot})$</td>
<td></td>
<td></td>
<td>$\emptyset$</td>
</tr>
</tbody>
</table>

- System response: $X = \text{spot}$ ?

- If we type “;” after the ? prompt (i.e., we ask for another solution) the system can go and execute a different branch (i.e., a different choice in $C_4^*$, or $C_1^*$).
The Search Tree Revisited

- Different execution strategies explore the tree in a different way.
- A strategy is complete if it guarantees that it will find all existing solutions.
The Search Tree Revisited

- Different execution strategies explore the tree in a different way.
- A strategy is complete if it guarantees that it will find all existing solutions.
- Standard Prolog does it top-down, left-to-right (i.e., depth-first).

```
pet(X) :- animal(X), barks(X).  animal(tim).  barks(spot).
pet(X) :- animal(X), meows(X).  animal(spot).  animal(hobbes).
```

```
X=tim
  animal(X), barks(X)
    X=spot
      barks(spot)  solution
    X=hobbes
      barks(hobbes)  fail
X=tim
  animal(X), meows(X)
    X=spot
      meows(spot)  solution
    X=hobbes
      meows(hobbes)  fail
```
Characterization of The Search Tree

• All solutions are at finite depth in the tree.

• Failures can be at finite depth or, in some cases, be an infinite branch.
Depth-First Search (Backtracking)

- Incomplete: may fall through an infinite branch before finding all solutions.
- But very efficient: it can be implemented with a call stack, very similar to a traditional programming language.
Breadth-First Search

- Will find all solutions before falling through an infinite branch.
- But costly in terms of time and memory.
- Used in all the following examples (via Ciao’s bf package).
Selecting breadth-first or depth-first search

- In the Ciao system we can select the search rule using the packages mechanism.

- Files should start with the following line:
  - To execute in breadth-first mode:
    ```prolog
    :- module(_,_,['bf/bfall']).
    ```
  - To execute in depth-first mode:
    ```prolog
    :- module(_,_,[]).
    ```

See the part on Developing Programs with a Logic Programming System for more details on the particular system used in the course (Ciao).
Control of Search Size in Depth-First Search (Backtracking)

Conventional programs (no search) execute conventionally.

Programs with search: programmer has at least three ways of controlling search:

1. The ordering of literals in the body of a clause:

   - Profound effect on the size of the computation (in the limit, on termination).
     Compare executing \( p(X), q(X, Y) \) with executing \( q(X, Y), p(X) \) in:

     \[
     \begin{align*}
     p(X) & :\!- \quad X = 4. \\
     p(X) & :\!- \quad X = 5. \\
     q(X, Y) & :\!- \quad X = 1, \quad Y = a, \ldots \\
     q(X, Y) & :\!- \quad X = 2, \quad Y = b, \ldots \\
     q(X, Y) & :\!- \quad X = 4, \quad Y = c, \ldots \\
     q(X, Y) & :\!- \quad X = 4, \quad Y = d, \ldots 
     \end{align*}
     \]

     \( p(X), q(X, Y) \) is more efficient: execution of \( p/2 \) reduces the choices of \( q/2 \).

   - Note that optimal order depends on the variable instantiation mode:
     E.g., for \( q(X, d), p(X) \), this order is better than \( p(X), q(X, d) \).
2. The *ordering of clauses* in a predicate:

- Affects the order in which solutions are generated.
  
  E.g., in the previous example we get:
  \[ X = 4, Y = c \] as the first solution and \[ X = 4, Y = d \] as the second.

  If we reorder \( q/2 \):

  \[
  \begin{align*}
  p(X) & : - X = 4. \\
  p(X) & : - X = 5. \\
  q(X, Y) & : - X = 4, Y = d, \ldots \\
  q(X, Y) & : - X = 4, Y = c. \ldots \\
  q(X, Y) & : - X = 2, Y = b, \ldots \\
  q(X, Y) & : - X = 1, Y = a, \ldots 
  \end{align*}
  \]

  we get \[ X = 4, Y = d \] first and then \[ X = 4, Y = c \].

- If a subset of the solutions is requested, then clause order affects:
  
  ◦ the size of the computation,
  
  ◦ and, at the limit, termination!

  Else, little significance unless computation is infinite and/or *pruning* is used.

3. The *pruning operators* (e.g., “cut”), which cut choices dynamically –see later.
Role of Unification in Execution

- As mentioned before, unification used to access data and give values to variables.  
  
  Example: Consider query \( ?- \text{animal}(A), \text{named}(A,\text{Name}) \). with:
  \[
  \text{animal(dog(tim))}. \quad \text{named(dog(Name),Name)}.
  \]

- Also, unification is used to pass parameters in procedure calls and to return values upon procedure exit.

<table>
<thead>
<tr>
<th>Q</th>
<th>R</th>
<th>Clause</th>
<th>( \theta )</th>
</tr>
</thead>
<tbody>
<tr>
<td>pet(P)</td>
<td>pet(P)</td>
<td>C_1*</td>
<td>{ P=X_1 }</td>
</tr>
<tr>
<td>pet(X_1)</td>
<td>animal(X_1), barks(X_1)</td>
<td>C_3*</td>
<td>{ X_1=spot }</td>
</tr>
<tr>
<td>pet(spot)</td>
<td>barks(spot)</td>
<td>C_6</td>
<td>{}</td>
</tr>
<tr>
<td>pet(spot)</td>
<td>—</td>
<td>—</td>
<td>—</td>
</tr>
</tbody>
</table>
“Modes”

- In fact, argument positions are not fixed a priori to be input or output.

*Example:* Consider query `?- pet(spot).` vs. `?- pet(X).`

or `?- plus( s(0), s(s(0)), Z).` vs. `?- plus( s(0), Y, s(s(s(0))))`. % Adds

- Thus, procedures can be used in different **modes** s.t. different sets of arguments are input or output in each mode.

- We sometimes use `+` and `-` to refer to, respectively, and argument being an input or an output, e.g.:

  `plus(+X, +Y, -Z)` means we call `plus` with
  - `X` instantiated,
  - `Y` instantiated, and
  - `Z` free.
Computational Logic

Pure Logic Programming Examples
Database Programming

- A Logic Database is a set of facts and rules (i.e., a logic program):

  father_of(john, peter).
  father_of(john, mary).
  father_of(peter, michael).
  mother_of(mary, david).

  grandfather_of(L, M) :- father_of(L, N),
                          father_of(N, M).
  grandfather_of(X, Y) :- father_of(X, Z),
                          mother_of(Z, Y).

- Given such database, a logic programming system can answer questions (queries) such as:

  ?- father_of(john, peter).
  yes
  ?- father_of(john, david).
  no
  ?- father_of(john, X).
  X = peter ;
  X = mary

- Rules for grandmother_of(X, Y)?

  ?- grandfather_of(X, michael).
  X = john
  ?- grandfather_of(X, Y).
  X = john, Y = michael ;
  X = john, Y = david
  ?- grandfather_of(X, X).
  no
Another example:

resistor(power, n1).
resistor(power, n2).

transistor(n2, ground, n1).
transistor(n3, n4, n2).
transistor(n5, ground, n4).

inverter(Input, Output) :-
    transistor(Input, ground, Output), resistor(power, Output).

nand_gate(Input1, Input2, Output) :-
    transistor(Input1, X, Output), transistor(Input2, ground, X),
    resistor(power, Output).

and_gate(Input1, Input2, Output) :-
    nand_gate(Input1, Input2, X), inverter(X, Output).

Query  ?- and_gate(In1, In2, Out)  has solution:  In1=n3, In2=n5, Out=n1
Structured Data and Data Abstraction (and the ’=’ Predicate)

- **Data structures** are created using (complex) terms.

- Structuring data is important:
  
  ```prolog
  course(complog,wed,18,30,20,30,'M.','Hermenegildo',new,5102).
  ```

- When is the Computational Logic course?
  
  ```prolog
  ```

- Structured version:
  
  ```prolog
  course(complog,Time,Lecturer, Location) :-
      Time = t(wed,18:30,20:30),
      Lecturer = lect('M.','Hermenegildo'),
      Location = loc(new,5102).
  ```

  **Note:** “X=Y” is equivalent to “’=’(X,Y)” where the predicate =/2 is defined as the fact “’=’(X,X).” – Plain unification!

- Equivalent to:
  
  ```prolog
  course(complog, t(wed,18:30,20:30),
      lect('M.','Hermenegildo'), loc(new,5102)).
  ```
Structured Data and Data Abstraction (and The Anonymous Variable)

- Given:

  \[
  \text{course(complog,Time,Lecturer, Location) :-}
  \]
  \[
  \quad \text{Time} = t(\text{wed},18:30,20:30),
  \]
  \[
  \quad \text{Lecturer} = \text{lect('M.','Hermenegildo')},
  \]
  \[
  \quad \text{Location} = \text{loc(new,5102)}.
  \]

- When is the Computational Logic course?

  ```
  ?- course(complog, Time, A, B).
  ```

  has solution:

  ```
  Time=t(wed,18:30,20:30), A=lect('M.','Hermenegildo'), B=loc(new,5102)
  ```

- Using the \textit{anonymous variable} ("_"):

  ```
  ?- course(complog,Time, _, _).
  ```

  has solution:

  ```
  Time=t(wed,18:30,20:30)
  ```
Terms as Data Structures with Pointers

- **main** below is a procedure, that:
  - creates some data structures, with *pointers* and *aliasing*.
  - *calls* other *procedures*, *passing* to them *pointers* to these structures.

```prolog
main :-
    X = f(K, g(K)),
    Y = a,
    Z = g(L),
    W = h(b, L),
    % Heap memory at this point →
    p(X,Y),
    q(Y,Z),
    r(W).
```

- Terms are data structures with pointers.
- Logical variables are *declarative* pointers.
  - Declarative: they can only be assigned once.
Structured Data and Data Abstraction (Contd.)

• The circuit example revisited:

```prolog
resistor(r1,power,n1).  transistor(t1,n2,ground,n1).
resistor(r2,power,n2).  transistor(t2,n3,n4,n2).
transistor(t3,n5,ground,n4).

inverter(inv(T,R),Input,Output) :-
  transistor(T,Input,ground,Output),
  resistor(R,power,Output).

nand_gate(nand(T1,T2,R),Input1,Input2,Output) :-
  transistor(T1,Input1,X,Output),
  transistor(T2,Input2,ground,X),
  resistor(R,power,Output).

and_gate(and(N,I),Input1,Input2,Output) :-
  nand_gate(N,Input1,Input2,X),
  inverter(I,X,Output).
```

• The query

```
?- and_gate(G,In1,In2,Out).
```

has solution:

```
G=and(nand(t2,t3,r2),inv(t1,r1)), In1=n3, In2=n5, Out=n1
```
Logic Programs and the Relational DB Model

<table>
<thead>
<tr>
<th>Relational Database</th>
<th>Logic Programming</th>
</tr>
</thead>
<tbody>
<tr>
<td>Relation Name</td>
<td>Predicate symbol</td>
</tr>
<tr>
<td>Relation</td>
<td>Procedure consisting of ground facts (facts without variables)</td>
</tr>
<tr>
<td>Tuple</td>
<td>Ground fact</td>
</tr>
<tr>
<td>Attribute</td>
<td>Argument of predicate</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Name</th>
<th>Age</th>
<th>Sex</th>
</tr>
</thead>
<tbody>
<tr>
<td>Brown</td>
<td>20</td>
<td>M</td>
</tr>
<tr>
<td>Jones</td>
<td>21</td>
<td>F</td>
</tr>
<tr>
<td>Smith</td>
<td>36</td>
<td>M</td>
</tr>
</tbody>
</table>

“Person”

<table>
<thead>
<tr>
<th>Name</th>
<th>Town</th>
<th>Years</th>
</tr>
</thead>
<tbody>
<tr>
<td>Brown</td>
<td>London</td>
<td>15</td>
</tr>
<tr>
<td>Brown</td>
<td>York</td>
<td>5</td>
</tr>
<tr>
<td>Jones</td>
<td>Paris</td>
<td>21</td>
</tr>
<tr>
<td>Smith</td>
<td>Brussels</td>
<td>15</td>
</tr>
<tr>
<td>Smith</td>
<td>Santander</td>
<td>5</td>
</tr>
</tbody>
</table>

“Lived in”

<table>
<thead>
<tr>
<th>Name</th>
<th>Town</th>
<th>Years</th>
</tr>
</thead>
<tbody>
<tr>
<td>Brown</td>
<td>London</td>
<td>15</td>
</tr>
<tr>
<td>Brown</td>
<td>York</td>
<td>5</td>
</tr>
<tr>
<td>Jones</td>
<td>Paris</td>
<td>21</td>
</tr>
<tr>
<td>Smith</td>
<td>Brussels</td>
<td>15</td>
</tr>
<tr>
<td>Smith</td>
<td>Santander</td>
<td>5</td>
</tr>
</tbody>
</table>

person(brown, 20, male).
person(jones, 21, female).
person(smith, 36, male).
lived_in(brown, london, 15).
lived_in(brown, york, 5).
lived_in(jones, paris, 21).
lived_in(smith, brussels, 15).
lived_in(smith, santander, 5).
Logic Programs and the Relational DB Model (Contd.)

- The operations of the relational model are easily implemented as rules.
  
  ◦ **Union:** \( r_{\text{union}} s(X_1, \ldots, X_n) \leftarrow r(X_1, \ldots, X_n). \)
  
    \( r_{\text{union}} s(X_1, \ldots, X_n) \leftarrow s(X_1, \ldots, X_n). \)
  
  ◦ **Cartesian Product:**
    
    \( r \times s(X_1, \ldots, X_m, X_{m+1}, \ldots, X_{m+n}) \leftarrow r(X_1, \ldots, X_m), s(X_{m+1}, \ldots, X_{m+n}). \)
  
  ◦ **Projection:** \( r_{13}(X_1, X_3) \leftarrow r(X_1, X_2, X_3). \)
  
  ◦ **Selection:** \( r_{\text{selected}}(X_1, X_2, X_3) \leftarrow r(X_1, X_2, X_3), \leq(X_2, X_3). \)
    
    \((\leq/2 \text{ can be, e.g., Peano, Prolog built-in, constraints...})\)
  
  ◦ **Set Difference:**
    
    \( r_{\text{diff}} s(X_1, \ldots, X_n) \leftarrow r(X_1, \ldots, X_n), \text{not } s(X_1, \ldots, X_n). \)
    
    \( r_{\text{diff}} s(X_1, \ldots, X_n) \leftarrow s(X_1, \ldots, X_n), \text{not } r(X_1, \ldots, X_n). \)
    
    \((\text{we postpone the discussion on negation until later.})\)
  
- Derived operations – some can be expressed more directly in LP:
  
  ◦ **Intersection:** \( r_{\text{meet}} s(X_1, \ldots, X_n) \leftarrow r(X_1, \ldots, X_n), s(X_1, \ldots, X_n). \)
  
  ◦ **Join:** \( r_{\text{joinX2}} s(X_1, \ldots, X_n) \leftarrow r(X_1, X_2, X_3, \ldots, X_n), s(X_1', X_2, X_3', \ldots, X_n'). \)
  
- Duplicates an issue: see “setof” later in Prolog.
The subject of “deductive databases” uses these ideas to develop *logic-based databases*.  

- Often syntactic restrictions (a subset of definite programs) used (e.g. “Datalog” – no functors, no existential variables).
- Variations of a “bottom-up” execution strategy used: Use the $T_p$ operator (explained in the theory part) to compute the model, restrict to the query.
- Powerful notions of negation supported: S-models
  → **Answer Set Programming** (ASP)
  → powerful knowledge representation and reasoning systems.
Recursive Programming

- Example: ancestors.

```prolog
parent(X, Y) :- father(X, Y).
parent(X, Y) :- mother(X, Y).

ancestor(X, Y) :- parent(X, Y).
ancestor(X, Y) :- parent(X, Z), ancestor(Z, Y).
ancestor(X, Y) :- parent(X, Z), parent(Z, W), ancestor(W, Y).
ancestor(X, Y) :- parent(X, Z), parent(Z, W), parent(W, K), ancestor(K, Y).
...```

- Defining ancestor recursively:

```prolog
parent(X, Y) :- father(X, Y).
parent(X, Y) :- mother(X, Y).

ancestor(X, Y) :- parent(X, Y).
ancestor(X, Y) :- parent(X, Z), ancestor(Z, Y).

Exercise: define “related”, “cousin”, “same generation”, etc.
Types

- **Type**: a (possibly infinite) set of terms.
- **Type definition**: A program defining a type.
- **Example**: Weekday:
  - Set of terms to represent: 'Monday', 'Tuesday', 'Wednesday', ...
  - Type definition:
    - weekday('Monday').
    - weekday('Tuesday'). ...

- **Example**: Date (weekday * day in the month):
  - Set of terms to represent: date('Monday', 23), date('Tuesday', 24), ...
  - Type definition:
    - date(date(W,D)) :- weekday(W), day_of_month(D).
    - day_of_month(1).
    - day_of_month(2).
    - ...
    - day_of_month(31).
Recursive Programming: Recursive Types

- **Recursive types**: defined by recursive logic programs.
- **Example**: natural numbers (simplest recursive data type):
  - Set of terms to represent: \(0, s(0), s(s(0)), \ldots\)
  - Type definition:

    \[
    \text{nat}(0). \\
    \text{nat}(s(X)) :- \text{nat}(X).
    \]

  A *minimal recursive predicate*:
  one unit clause and one recursive clause (with a single body literal).
- Types are *runnable* and can be used to check or produce values:
  - \(?- \text{nat}(X) \Rightarrow X=0; X=s(0); X=s(s(0)); \ldots\)
- We can reason about *complexity*, for a given class of queries ("mode").
  E.g., for mode \(\text{nat}(\text{ground})\) complexity is *linear* in size of number.
- **Example**: integers:
  - Set of terms to represent: \(0, s(0), -s(0), \ldots\)
  - Type definition:

    \[
    \text{integer}(X) :- \text{nat}(X). \\
    \text{integer}(-X) :- \text{nat}(X).
    \]
Recursive Programming: Arithmetic

- Defining the natural order ($\leq$) of natural numbers:

  \begin{verbatim}
  less_or_equal(0, X) :- nat(X).
  less_or_equal(s(X), s(Y)) :- less_or_equal(X, Y).
  \end{verbatim}

  ◦ Multiple uses (modes):

  \begin{verbatim}
  less_or_equal(s(0), s(s(0))), less_or_equal(X, 0), ...
  \end{verbatim}

  ◦ Multiple solutions:

  \begin{verbatim}
  less_or_equal(X, s(0)), less_or_equal(s(s(0)), Y), etc.
  \end{verbatim}

- Addition:

  \begin{verbatim}
  plus(0, X, X) :- nat(X).
  plus(s(X), Y, s(Z)) :- plus(X, Y, Z).
  \end{verbatim}

  ◦ Multiple uses (modes): \begin{verbatim}
  plus(s(s(0)), s(0), Z), plus(s(s(0)), Y, s(0))
  \end{verbatim}

  ◦ Multiple solutions: \begin{verbatim}
  plus(X, Y, s(s(s(0)))), etc.
  \end{verbatim}
Recursive Programming: Arithmetic

- Another possible definition of addition:

  \[
  \begin{align*}
  \text{plus}(X,0,X) & :\text{- nat}(X). \\
  \text{plus}(X,s(Y),s(Z)) & :\text{- plus}(X,Y,Z).
  \end{align*}
  \]

- The meaning of \texttt{plus} is the same if both definitions are combined.

- Not recommended: several proof trees for the same query \(\rightarrow\) not efficient, not concise. We look for minimal axiomatizations.

- The art of logic programming: finding compact and computationally efficient formulations!

- Try to define: \(\text{times}(X,Y,Z) \quad (Z = X \cdot Y)\), \(\text{exp}(N,X,Y) \quad (Y = X^N)\), \(\text{factorial}(N,F) \quad (F = N!)\), \(\text{minimum}(N1,N2,\text{Min})\), \(\ldots\)
Recursive Programming: Arithmetic

- Definition of \( \text{mod}(X, Y, Z) \)
  “Z is the remainder from dividing X by Y”

\[ \exists Q \text{ s.t. } X = Y \times Q + Z \land Z < Y \]

\[ \Rightarrow \]

\[ \text{mod}(X, Y, Z) :- \text{less}(Z, Y), \text{times}(Y, Q, W), \text{plus}(W, Z, X). \]

\[ \text{less}(\emptyset, s(X)) :- \text{nat}(X). \]
\[ \text{less}(s(X), s(Y)) :- \text{less}(X, Y). \]

- Another possible definition:

\[ \text{mod}(X, Y, X) :- \text{less}(X, Y). \]
\[ \text{mod}(X, Y, Z) :- \text{plus}(X1, Y, X), \text{mod}(X1, Y, Z). \]

- The second is much more efficient than the first one (compare the size of the proof trees).
The Ackermann function:

- \( \text{ackermann}(0, N) = N + 1 \)
- \( \text{ackermann}(M, 0) = \text{ackermann}(M - 1, 1) \)
- \( \text{ackermann}(M, N) = \text{ackermann}(M - 1, \text{ackermann}(M, N - 1)) \)

In Peano arithmetic:

- \( \text{ackermann}(0, N) = s(N) \)
- \( \text{ackermann}(s(M1), 0) = \text{ackermann}(M1, s(0)) \)
- \( \text{ackermann}(s(M1), s(N1)) = \text{ackermann}(M1, \text{ackermann}(s(M1), N1)) \)

Can be defined as:

- \( \text{ackermann}(0, N, s(N)) \).
- \( \text{ackermann}(s(M1), 0, Val) :- \text{ackermann}(M1, s(0), Val) \).
- \( \text{ackermann}(s(M1), s(N1), Val) :- \text{ackermann}(s(M1), N1, Val1), \text{ackermann}(M1, Val1, Val) \).

I.e., in general, functions can be coded as a predicate with one more argument, which represents the output (and additional syntactic sugar often available).
• **:- use_package(fsyntax).** Provides:

- `~` "eval", which makes the last argument implicit. This allows writing, e.g.
  
  ```prolog
  p(X, Y) :- q(X, Z), r(Z, Y).
  
  as
  p(X, Y) :- r(~q(X), Y).
  
  or
  p(X, ~r(~q(X)).
  ```

- `:=` for definitions: which allows writing, e.g.
  
  ```prolog
  p(X, Y) :- q(X, Z), r(Z, Y).
  
  as
  p(X) := Y :- r(~q(X), Y).
  
  or
  p(X) := ~r(~q(X)).
  ```

- `|` for or, etc.
Thus, we can now write:

\[
\text{ackmann}(s\ M, s\ N) := \neg\text{ackmann}(M, \neg\text{ackmann}(s\ M, N)).
\]

To evaluate automatically functors that are defined as functions (so there is no need to use \(\neg\) for them):

\[
\text{:- fun_eval ackmann/2.}
\]

\[
\text{ackmann}(s\ M, s\ N) := \text{ackmann}(M, \text{ackmann}(s\ M, N)).
\]

To enable this for all functions defined in a given file:

\[
\text{:- fun_eval defined(true).}
\]

To evaluate arithmetic functors automatically (no need for \(\neg\) for them):

\[
\text{:- fun_eval arith(true).}
\]

\[
\text{add_one(X,X+1).}
\]

The functional package includes fsyntax and both fun_eval’s above:

\[
\text{:- use_package(functional).}
\]
The Ackermann function (Peano) in Ciao’s functional Syntax and defining s as a prefix operator:

```prolog
:- use_package(functional).
:- op(500,fy,s).

ackermann( 0, N) := s N.
ackermann(s M, 0) := ackermann(M, s 0).
ackermann(s M, s N) := ackermann(M, ackermann(s M, N) ).
```

Convenient in other cases – e.g. for defining types:

```prolog
nat(0).
nat(s(X)) :- nat(X).
```

Using special := notation for the “return” (last) the argument:

```prolog
nat := 0.
nat := s(X) :- nat(X).
```
Recursive Programming: Arithmetic/Functions (Funct. Syntax, Contd.)

Moving body call to head using the ~ notation ("evaluate and replace with result"):

\[ \text{nat} := 0. \]
\[ \text{nat} := s(\text{~nat}). \]

"~" not needed with functional package if inside its own definition:

\[ \text{nat} := 0. \]
\[ \text{nat} := s(\text{nat}). \]

Using an \texttt{:- op(500,fy,s)} declaration to define \texttt{s} as a prefix operator:

\[ \text{nat} := 0. \]
\[ \text{nat} := s \text{ nat}. \]

Using "|" (disjunction):

\[ \text{nat} := 0 | s \text{ nat}. \]

Which is exactly equivalent to:

\[ \text{nat}(0). \]
\[ \text{nat}(s(X)) :- \text{nat}(X). \]
Recursive Programming: Lists

- Binary structure: first argument is *element*, second argument is *rest* of the list.

- We need:
  - A constant symbol: we use the *constant* `[ ]` (→ denotes the empty list).
  - A functor of arity 2: traditionally the dot “.” (which is overloaded).

- Syntactic sugar: the term `.X,Y` is denoted by `[X|Y]` (X is the *head*, Y is the *tail*).

<table>
<thead>
<tr>
<th>Formal object</th>
<th>“Cons pair” syntax</th>
<th>“Element” syntax</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>(a,[])</code></td>
<td>`[a</td>
<td>[[]]]`</td>
</tr>
<tr>
<td><code>(a,(b,[]))</code></td>
<td>`[a,[b</td>
<td>[[]]]]`</td>
</tr>
<tr>
<td><code>(a,(b,(c,[])))</code></td>
<td>`[a,[b,[c</td>
<td>[[]]]]]`</td>
</tr>
<tr>
<td><code>(a,X)</code></td>
<td>`[a</td>
<td>X]`</td>
</tr>
<tr>
<td><code>(a,(b,X))</code></td>
<td>`[a,[b</td>
<td>X]]`</td>
</tr>
</tbody>
</table>

- Note that:
  - `[a,b]` and `[a|X]` unify with `{X = [b]}`
  - `[a]` and `[a|X]` unify with `{X = []}`
  - `[a]` and `[a,b|X]` do not unify
  - `[]` and `[X]` do not unify
Recursive Programming: Lists (Contd.)

- Type definition (no syntactic sugar):
  
  ```prolog
  list([]).
  list([X,Y]) :- list(Y).
  ```

- Type definition, with some syntactic sugar ([ ] notation):
  
  ```prolog
  list([]).
  list([X|Y]) :- list(Y).
  ```

- Type definition, using also functional package:
  
  ```prolog
  list :- [] | [_|list].
  ```

- “Exploring” the type:
  
  ```prolog
  %- list(L).
  L = [] ? ;
  L = [_] ? ;
  L = [_,_] ? ;
  L = [_,_,_] ?
  ...
  ```
• X is a *member* of the list Y:

<table>
<thead>
<tr>
<th>member(a,[a]).</th>
<th>member(b,[b]).</th>
<th>etc.  (\Rightarrow) member(X,[X]).</th>
</tr>
</thead>
<tbody>
<tr>
<td>member(a,[a,c]).</td>
<td>member(b,[b,d]).</td>
<td>etc.  (\Rightarrow) member(X,[X,Y]).</td>
</tr>
<tr>
<td>member(a,[a,c,d]).</td>
<td>member(b,[b,d,l]).</td>
<td>etc.  (\Rightarrow) member(X,[X,Y,Z]).</td>
</tr>
</tbody>
</table>

\[\Rightarrow \text{member}(X,[X|Y]) \leftarrow \text{list}(Y).\]

<table>
<thead>
<tr>
<th>member(a,[c,a]),</th>
<th>member(b,[d,b]).</th>
<th>etc.  (\Rightarrow) member(X,[Y,X]).</th>
</tr>
</thead>
<tbody>
<tr>
<td>member(a,[c,d,a]).</td>
<td>member(b,[s,t,b]).</td>
<td>etc.  (\Rightarrow) member(X,[Y,Z,X]).</td>
</tr>
</tbody>
</table>

\[\Rightarrow \text{member}(X,[Y|Z]) \leftarrow \text{member}(X,Z).\]

• Resulting definition:

| member(X,[X|Y]) \leftarrow list(Y). |
| member(X,[_|T]) \leftarrow member(X,T). |

• Uses of member(X,Y):
  ◦ checking whether an element is in a list (\text{member}(b,[a,b,c]))
  ◦ finding an element in a list (\text{member}(X,[a,b,c]))
  ◦ finding a list containing an element (\text{member}(a,Y))
• Combining lists and naturals:

   ◇ Computing the length of a list:

   ```prolog
   len([], 0).
   len([H|T], s(LT)) :- len(T, LT).
   ```

   ◇ Adding all elements of a list:

   ```prolog
   sumlist([], 0).
   sumlist([H|T], S) :- sumlist(T, ST), plus(ST, H, S).
   ```

   ◇ The type of lists of natural numbers:

   ```prolog
   natlist([]).
   natlist([H|T]) :- nat(H), natlist(T).
   ```

   or:

   ```prolog
   natlist := [] | [~nat|natlist].
   ```
Exercises:

- Define: \texttt{prefix(X,Y)} (the list \texttt{X} is a prefix of the list \texttt{Y}), e.g. 
  \texttt{prefix([a, b], [a, b, c, d])}
- Define: \texttt{suffix(X,Y)}, \texttt{sublist(X,Y)},...
Recursive Programming: Lists (Contd.)

- Concatenation of lists:
  - Base case:
    append([], [a], [a]). append([], [a, b], [a, b]). etc.
    \[\Rightarrow\] append([], Ys, Ys) :- list(Ys).

  - Rest of cases (first step):
    append([a], [b], [a, b]).
    append([a], [b, c], [a, b, c]). etc.
    \[\Rightarrow\] append([X], Ys, [X|Ys]) :- list(Ys).
    append([a, b], [c], [a, b, c]).
    append([a, b], [c, d], [a, b, c, d]). etc.
    \[\Rightarrow\] append([X, Z], Ys, [X, Z|Ys]) :- list(Ys).

This is still infinite \[\Rightarrow\] we need to generalize more.
Recursive Programming: Lists (Contd.)

- Second generalization:
  
  \[
  \text{append}([X], Ys, [X|Ys]) :- \text{list}(Ys).
  \]

  \[
  \text{append}([X,Z], Ys, [X,Z|Ys]) :- \text{list}(Ys).
  \]

  \[
  \text{append}([X,Z,W], Ys, [X,Z,W|Ys]) :- \text{list}(Ys).
  \]

  \[
  \implies \text{append}([X|Xs], Ys, [X|Zs]) :- \text{append}(Xs, Ys, Zs).
  \]

- So, we have:
  
  \[
  \text{append}([], Ys, Ys) :- \text{list}(Ys).
  \]

  \[
  \text{append}([X|Xs], Ys, [X|Zs]) :- \text{append}(Xs, Ys, Zs).
  \]

- Another way of reasoning: thinking inductively.
  
  ◇ The base case is: \text{append}([], Ys, Ys) :- \text{list}(Ys).

  ◇ If we assume that \text{append}(Xs, Ys, Zs) works for some iteration, then, in the next one, the following holds: \text{append}([X|Xs], Ys, [X|Zs]).
Recursive Programming: Lists (Contd.)

- Uses of append:
  - Concatenate two given lists:
    ```prolog
    ?- append([a, b, c], [d, e], L).
    L = [a, b, c, d, e].
    ```
  - Find differences between lists:
    ```prolog
    ?- append(D, [d, e], [a, b, c, d, e]).
    D = [a, b, c].
    ```
  - Split a list:
    ```prolog
    ?- append(A, B, [a, b, c, d, e]).
    A = [],
    B = [a, b, c, d, e].
    A = [a],
    B = [b, c, d, e].
    A = [a, b],
    B = [c, d, e].
    A = [a, b, c],
    B = [d, e].
    ...
    ```
Recursive Programming: Lists (Contd.)

- \text{reverse}(Xs, Ys): Ys is the list obtained by reversing the elements in the list Xs.
  
  Each element X of [X|Xs] should end up at the end of the reversed version of Xs:

  \[
  \text{reverse}([X|Xs], Ys) ::= \\
  \text{reverse}(Xs, Zs), \\
  \text{append}(Zs, [X], Ys). 
  \]

  Inductively: if we assume Xs is already reversed as Zs, if Xs has one more element at the beginning, it goes at the end of Zs.

  How can we stop (i.e., what is our base case):
  \[
  \text{reverse}([], []). 
  \]

- As defined, \text{reverse}(Xs,Ys) is very inefficient. Another possible definition:
  (uses an \textit{accumulating parameter})

  \[
  \text{reverse}(Xs, Ys) ::= \text{reverse}(Xs, [], Ys). \\
  \text{reverse}([], Ys, Ys). \\
  \text{reverse}([X|Xs], Acc, Ys) ::= \text{reverse}(Xs, [X|Acc], Ys). 
  \]

  \Rightarrow \text{Find the differences in terms of efficiency between the two definitions.}
Recursive Programming: Binary Trees

- Represented by a ternary functor \( \text{tree}(\text{Element}, \text{Left}, \text{Right}) \).
- Empty tree represented by \text{void}.
- Definition:

\[
\begin{align*}
\text{binary_tree}(\text{void}). \\
\text{binary_tree}(\text{tree}(\_\text{Element}, \text{Left}, \text{Right})) & : - \\
& \quad \text{binary_tree}(\text{Left}), \\
& \quad \text{binary_tree}(\text{Right}).
\end{align*}
\]

- Defining \text{tree_member}(\text{Element}, \text{Tree}):

\[
\begin{align*}
\text{tree_member}(X, \text{tree}(X, \text{Left}, \text{Right})) & : - \\
& \quad \text{binary_tree}(\text{Left}), \\
& \quad \text{binary_tree}(\text{Right}). \\
\text{tree_member}(X, \text{tree}(\_, \text{Left}, \text{Right})) & : - \text{tree_member}(X, \text{Left}). \\
\text{tree_member}(X, \text{tree}(\_, \text{Left}, \text{Right})) & : - \text{tree_member}(X, \text{Right}).
\end{align*}
\]
Recursive Programming: Binary Trees

- **Defining pre_order(Tree,Elements):**
  Elements is a list containing the elements of Tree traversed in *preorder*.

```
pre_order(void, []).
pre_order(tree(X, Left, Right), Elements) :-
    pre_order(Left, ElementsLeft),
    pre_order(Right, ElementsRight),
    append([X | ElementsLeft], ElementsRight, Elements).
```

- **Exercise – define:**
  - in_order(Tree, Elements)
  - post_order(Tree, Elements)
Polymorphism

• Note that the two definitions of \texttt{member/2} can be used \textit{simultaneously}:

\begin{verbatim}
lt_member(X,[X|Y]) :- list(Y).
ltn_member(X,[_|T]) :- lt_member(X,T).

lt_member(X,tree(X,L,R)) :- binary_tree(L), binary_tree(R).
ltn_member(X,tree(Y,L,R)) :- lt_member(X,L).
ltn_member(X,tree(Y,L,R)) :- lt_member(X,R).
\end{verbatim}

Lists only unify with the first two clauses, trees with clauses 3–5!

• \begin{verbatim}
:- lt_member(X,[b,a,c]).
X = b ; X = a ; X = c
\end{verbatim}

• \begin{verbatim}
:- lt_member(X,tree(b,tree(a,void,void),tree(c,void,void))).
X = b ; X = a ; X = c
\end{verbatim}

• Also, try (somewhat surprising): \begin{verbatim}
:- lt_member(M,T).
\end{verbatim}
• Recognizing (and generating!) polynomials in some term X:
  ◦ X is a polynomial in X
  ◦ a constant is a polynomial in X
  ◦ sums, differences and products of polynomials in X are polynomials
  ◦ also polynomials raised to the power of a natural number and the quotient of a polynomial by a constant

```prolog
polynomial(X,X).  
polynomial(Term,X) :- pconstant(Term).  
polynomial(Term1+Term2,X) :- polynomial(Term1,X), polynomial(Term2,X).  
polynomial(Term1-Term2,X) :- polynomial(Term1,X), polynomial(Term2,X).  
polynomial(Term1*Term2,X) :- polynomial(Term1,X), polynomial(Term2,X).  
polynomial(Term1/Term2,X) :- polynomial(Term1,X), pconstant(Term2).  
polynomial(Term1^N,X) :- polynomial(Term1,X), nat(N).
```
Symbolic differentiation: deriv(Expression, X, Derivative)

\[
\begin{align*}
\text{deriv}(X,X,s(0)) &. \\
\text{deriv}(C,X,0) &: - pconstant(C). \\
\text{deriv}(U+V,X,DU+DV) &: - \text{deriv}(U,X,DU), \text{deriv}(V,X,DV). \\
\text{deriv}(U-V,X,DU-DV) &: - \text{deriv}(U,X,DU), \text{deriv}(V,X,DV). \\
\text{deriv}(U*V,X,DU*V+U*DV) &: - \text{deriv}(U,X,DU), \text{deriv}(V,X,DV). \\
\text{deriv}(U/V,X,(DU*V-U*DV)/V^s(s(0))) &: - \text{deriv}(U,X,DU), \text{deriv}(V,X,DV). \\
\text{deriv}(U^s(N),X,s(N)*U^N*DU) &: - \text{deriv}(U,X,DU), \text{nat}(N). \\
\text{deriv}(\log(U),X,DU/U) &: - \text{deriv}(U,X,DU). \\
\end{align*}
\]

?- deriv(s(s(s(0)))*x+s(s(0)),x,Y).

A simplification step can be added.
Recursive Programming: Graphs

- A common approach: make use of another data structure, e.g., lists:
  - Graphs as lists of edges.

- Alternative: make use of Prolog's program database:
  - Declare the graph using facts in the program.

```prolog
edge(a,b). edge(c,a).
edge(b,c). edge(d,a).
```

- Paths in a graph: `path(X,Y)` iff there is a path in the graph from node `X` to node `Y`.

```prolog
path(A,B) :- edge(A,B).
p{}ath(A,B) :- edge(A,X), path(X,B).
```

- Circuit: a closed path. `circuit` iff there is a path in the graph from a node to itself.

```prolog
circuit :- path(A,A).
```
Recursive Programming: Graphs (Exercises)

- Modify `circuit/0` so that it gives the circuit. (You have to modify also `path/2`)
- Propose a solution for handling several graphs in our representation.
- Propose a suitable representation of graphs as data structures.
- Define the previous predicates for your representation.

- Consider unconnected graphs (there is a subset of nodes not connected in any way to the rest) versus connected graphs.
- Consider directed versus undirected graphs.

- Try `path(a,d)`. Solve the problem.
Recursive Programming: Automata (Graphs)

- Recognizing the sequence of characters accepted by the following non-deterministic, finite automaton (NDFA):
  
  ![Diagram of NDFA](image)

  where q₀ is both the initial and the final state.

- Strings are represented as lists of constants (e.g., [a, b, b]).

- Program:

  ```prolog
  initial(q₀).
  delta(q₀,a,q₁).
  delta(q₁,b,q₀).
  final(q₀).
  delta(q₁,b,q₁).
  accept(S) :- initial(Q), accept_from(S,Q).
  accept_from([],Q) :- final(Q).
  accept_from([X|Xs],Q) :- delta(Q,X,NewQ), accept_from(Xs,NewQ).
  ```


A *nondeterministic, stack, finite automaton* (NDSFA):

\[
\text{accept}(S) :- \text{initial}(Q), \text{accept}._\text{from}(S,Q,[]).
\]

\[
\text{accept}._\text{from}([],Q,[]) :- \text{final}(Q).
\]

\[
\text{accept}._\text{from}([X|Xs],Q,S) :- \text{delta}(Q,X,S,NewQ,NewS), \text{accept}._\text{from}(Xs,NewQ,NewS).
\]

\[
\text{initial}(q\emptyset).
\]

\[
\text{final}(q1).
\]

\[
\text{delta}(q\emptyset,X,Xs,q\emptyset,[X|Xs]).
\]

\[
\text{delta}(q\emptyset,X,Xs,q1,[X|Xs]).
\]

\[
\text{delta}(q\emptyset,X,Xs,q1,Xs).
\]

\[
\text{delta}(q1,X,[X|Xs],q1,Xs).
\]

What sequence does it recognize?
Recursive Programming: Towers of Hanoi

- **Objective:**
  - Move tower of N disks from peg a to peg b, with the help of peg c.

- **Rules:**
  - Only one disk can be moved at a time.
  - A larger disk can never be placed on top of a smaller disk.

---

N = 1

a

b

b

N = 2

b

b

N = 3

b

b

b

b

b

b

N = 2

b

b

N = 3

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b

b
Recursive Programming: Towers of Hanoi (Contd.)

- We will call the main predicate `hanoi_moves(N, Moves)`
- \( N \) is the number of disks and \( \text{Moves} \) the corresponding list of “moves”.
- Each move `move(A, B)` represents that the top disk in A should be moved to B.
- **Example:**

```
move(a, b), move(a, c), move(b, c), move(a, b),
move(c, a), move(c, b), move(a, b)
```

is represented by:

```
``
Recursive Programming: Towers of Hanoi (Contd.)

- A general rule:

- We capture this in a predicate \( hanoi(N, \text{Orig}, \text{Dest}, \text{Help}, \text{Moves}) \) where "Moves contains the moves needed to move a tower of \( N \) disks from peg \( \text{Orig} \) to peg \( \text{Dest} \), with the help of peg \( \text{Help} \)."

\[
\begin{align*}
\text{hanoi}(s(0), \text{Orig}, \text{Dest}, \_\text{Help}, [\text{move}(\text{Orig}, \text{Dest})]). \\
\text{hanoi}(s(N), \text{Orig}, \text{Dest}, \text{Help}, \text{Moves}) & : - \\
& \quad \text{hanoi}(N, \text{Orig}, \text{Help}, \text{Dest}, \text{Moves1}), \\
& \quad \text{hanoi}(N, \text{Help}, \text{Dest}, \text{Orig}, \text{Moves2}), \\
& \quad \text{append}(\text{Moves1}, [\text{move}(\text{Orig}, \text{Dest})|\text{Moves2}], \text{Moves}).
\end{align*}
\]

- And we simply call this predicate:

\[
\begin{align*}
\text{hanoi\_moves}(N, \text{Moves}) & : - \\
& \quad \text{hanoi}(N, a, b, c, \text{Moves}).
\end{align*}
\]
Learning to Compose Recursive Programs

- To some extent it is a simple question of practice.
- By generalization (as in the previous examples): elegant, but sometimes difficult? (Not the way most people do it.)
- Think inductively: state first the base case(s), and then think about the general recursive case(s).
- Sometimes it may help to compose programs with a given use in mind (e.g., “forwards execution”), making sure it is declaratively correct. Consider then also if alternative uses make sense.
- Sometimes it helps to look at well-written examples and use the same “schemas.”
- Using a global top-down design approach can help (in general, not just for recursive programs):
  - State the general problem.
  - Break it down into subproblems.
  - Solve the pieces.
- Again, the best approach: practice, practice, practice.